
;#define     dtr    PORTx, Pinx    ; Must specify which Port and Which Pin 
;#define     cts    PORTx, Pinx    ; Must specify which Port and Which Pin 
;#define     rts    PORTx, Pinx    ; Must specify which Port and Which Pin 
;#define     clr    PORTx, Pinx    ; Must specify which Port and Which Pin 
; 
;***************************************************************** 
; String Table 
; This table stores a string, breg is the offset.  The string 
; is terminated by a null character. 
;***************************************************************** 
string1 clrf   PCLATH           ; this routine is on page 0 
        movf   breg, W          ; get the offset 
        addwf  PCL, F           ; add the offset to PC 
        DT     D'15'            ; the first byte is the byte count 
        DT     "My IR ID String" 
; 
UpdateID 
        call   deviceInit       ; Initialize the PIC16Fxxx 
        bcf    clr              ; place the MCP2155 in reset 
        bsf    dtr              ; Force the DTR pin High for program mode  
        bcf    rts              ; Force the RTS pin Low for program mode 
        call   delay1mS         ; delay for 1 ms. 
        bsf    clr              ; allow the MCP2155 to come out of reset 
; 
        clrf   LoopCnt          ; LoopCnt = 0 
ctsLP1  call   delay1mS         ; delay for 1 ms. 
        btfss  cts              ; if cts=0 then we're ready to program 
        goto   ctsLow           ; MCP2155 is ready to receive data 
        decfsz LoopCnt, F       ; 
        goto   ctsLP1           ; NO, wait for MCP2155 to be ready 
        goto   StuckReset       ; The MCP2150 did not exit reset, do your recovery
                                ;   in this routine. 

ctsLow  clrf    breg            ; clear the offset 
        call    string1         ; get the byte count 
                                ;   (ID length byte + # bytes in string) 
        movwf   creg            ; use creg as the loop counter 
        incf    creg, f         ; add 1 to the loop count since 
                                ;    we're jumping into the middle 
        movwf   areg            ; save the count in areg to send it 
        goto    sndwt           ; start sending the count + ID string 
; 
sndlp   call    string1         ; get the byte 
        movwf   areg            ; save the byte 
sndwt   btfsc   cts             ; check the cts input 
        goto    sndwt           ; wait if cts=1 
        call    txser           ; send the byte using the Transmit Routine 
        incf    breg,f          ; increment the table pointer 
        decfsz  creg, f         ; more bytes to send? 
        goto    sndlp           ; YES, send more bytes 
; 
        bcf     clr             ; NO, place the MCP2155 in reset  
        bcf     dtr             ; Force the DTR pin Low for normal mode 
        bsf     rts             ; Force the RTS pin High for normal mode 
        call    delay1mS        ; delay for 1 ms. 
        bsf     clr             ; allow the MCP2155 to come out of reset 
; 
ctsLP2  btfss   cts             ; if cts=1 then MCP2155 is in Normal mode 
        goto    ctsLP2          ; NO, wait for MCP2155 to be ready 
        goto    NormalOperation ; The MCP2155 in now programmed with new ID, 
                                ; and is ready to establish an IR link 

